IDENTIFICATION

Product Code: DEC-08-YPPA-D
(previously, DIGITAL-8-6-U)

Product Name: Octal Memory Dump
(Octal Core Dump to Paper Tape)

Date: November 12, 1969

Maintenance
Contact: Software Information Service (CDB)

OCTAL MEMORY DUMP

Octal Core Dump to Paper Tape

1. ABSTRACT

This program enables the user to dump in octal mode any or all data in any memory field to either the Teletype or high-speed paper tape punch. During dumping, the absolute address of each location being dumped is held in the accumulator. When dumping is completed, output devices and memory fields may be changed to dump another section of memory.

2. REQUIREMENTS

2.1 Storage

This program requires one core page; initially 7400-7577.

2.2 Equipment

Any PDP-8 family computer with at least 4K words of core, an ASR-33 Teletype, and/or high-speed paper tape punch.

2.3 Software

No additional software is required. The program leaves the BIN and RIM Loaders untouched.

3. USAGE

The program is supplied in ASCII format on punched paper tape, and may be assembled by any 4K PDP-8 assembler, viz., PAL III, MACRO-8, PAL-D. The origin of this program (7400) may be changed with the PDP-8 Symbolic Editor in order to dump locations 7400-7577. (See the appropriate assembler manual for assembly instructions.)

3.1 Loading

The program is loaded into core with the Binary Loader (see DEC-08-LBAA-D or DEC-08-NGCC-D for loading procedures), and may be loaded into any available memory field.

4. OPERATING PROCEDURES

The switch register on the PDP-8 console is used to control the program. All options are taken from the position of bit Ø. The program may be interrupted by depressing the STOP switch.

With Memory Dump program in core:

1. Set the starting address and data field in the switch register and press the LOAD ADDress switch.
2. Set switch register bit Ø to 1 for a core dump to the Teletype punch, or to Ø when dumping via the high-speed paper tape punch.
3. Press the START switch. The computer will halt.
4. Set the switch register to the starting address of the section of core to be dumped.
5. Press the CONTinue switch. The computer will halt.
6. Set the switch register to the final core address of the section of core to be dumped.
7. Press the CONTinue switch; dumping commences and stops after dumping the contents of the final core address specified in step 6 above.

Another dumping session may be performed at this time by continuing at step 1 when you desire to change the output device or data field. Otherwise, continue at step 4.

The program will halt after each dumping session.

The preceding procedures are illustrated in Figure 1. on the next page.

5. ERROR MESSAGES

There are no error diagnostics or messages.

6. EXECUTION TIME

Execution (run) time is dependent on the amount of core dumped and the device used.

7. INPUT/OUTPUT

The program contains its own Teletype and high-speed punch output, and there are no external I/O handlers used. Switch register bit Ø determines the output device.



Figure 1. Operating Procedures.

8. FUNCTIONAL DESCRIPTION

The program is written in the PAL III language. Four routines are used in the program:

1. The TOCT routine causes a number to be formatted for a typeout or punchout.

2. The TCR routine outputs a carriage return-line feed.

3. The TSP routine outputs a space.

4. The TCHAR routine is the output routine for both the Teletype and high-speed punch.

The main routine begins with the initialization of variables, and the two address arguments are picked up from the switch register. Two carriage return-line feeds are performed, followed by the starting address and several spaces. A loop is then entered to type the contents of eight memory locations (if eight remain). If more data remains to be output, a JMP to LPØ2 repeats the process. If during this loop the routine finds that it has processed the last memory location, the loop exits, a carriage return-line feed is performed, a JMP to LPØØ is executed, and the program halts.

See the program listing that follows for more precise information.

9. PROGRAM LISTING

A printout of the program listing is furnished below.

```
                / OCTAL MEMORY DUMP PROGRAM
                / COPYRIGHT 1969
                / DIGITAL EQUIPMENT CORP.
                / MAYNARD, MASS.
                /TO OPERATE:
                /       LOAD ADDRESS 7400 IN SR
                /       TO CHOSE OUTPUT DEVICE:
                /       SET BIT 0=0 FOR H. S. PUNCH OUTPUT OR
                /       SET BIT 0=1 FOR TTY OUTPUT THEN PRESS START
                /       SET STARTING ADDRESS AND DATA FIELD IN SR -PRESS CONTINUE
                /       SET ENDING ADDRESS AND DATA FIELD IN SR -PRESS CONTINUE
                        *7400
7400    7604    DUMP,   CLA OSR         /EXAMINE SR FOR OUTPUT DEVICE
7401    7700            SMA CLA
7402    1265            TAD C10
7403    1270            TAD C7400
7404    3325            DCA SKPZ        /STORE A "SKP" IN SKPZ IF H. S. PUNCH OUTPUT
7405    7402    LP00,   HLT             /STOP. ENTER DUMP STARTING ADDRESS
7406    7604            LAS
7407    3261            DCA ADDR
7410    7402            HLT             /STOP. ENTER DUMP ENDING ADDRESS
7411    7604            LAS
7412    7040            CMA
7413    1261            TAD ADDR
7414    3262            DCA INDEX       /COUNTER FOR NUM OF LOCS TO BE DUMPED
7415    4312            JMS TCR         /TYPE CR-LFS
7416    4312    LP01,   JMS TCR
7417    1261            TAD ADDR
7420    4272            JMS TOCT        /OUTPUT STARTING ADDRESS IN OCTAL
```

```
7421  4320            JMS TSP         /OUTPUT 3 SPACES
7422  4320            JMS TSP
7423  4320  LP02,     JMS TSP
7424  1661            TAD I ADDR      /GET CONTENTS OF LOC
7425  4272            JMS TOCT        /TYPE OUT CONTENTS
7426  2262            ISZ INDEX       /DONE DUMPING?
7427  7410            SKP
7430  5247            JMP OUT         /YES. EXIT
7431  2261            ISZ ADDR        /NO. KICK ADDRESS UP
7432  1261            TAD ADDR        /HAVE WE OUTPUT 8 LOCS ON A LINE?
7433  0263            AND C3
7434  7640            SZA CLA
7435  5223            JMP LP02        /NO. SPACE OVER ONE AND GET NEXT
7436  1261            TAD ADDR
7437  0264            AND C7
7440  7640            SZA CLA
7441  5222            JMP LP02-1
7442  1261            TAD ADDR
7443  0266            AND C177
7444  7640            SZA CLA
7445  5216            JMP LP01        /OUTPUT CR/LF THEN NEW ADDRESS
7446  5215            JMP LP01-1
7447  4312  OUT,      JMS TCR         /OUTPUT CR/LF
7450  1267            TAD C214
7451  4324            JMS TCHAR       /OUTPUT A FORM FEED
7452  1271            TAD M20         /THEN OUTPUT 20 BLANKS OF TRAILER
7453  3262            DCA INDEX
7454  4324            JMS TCHAR
7455  2262            ISZ INDEX
7456  5254            JMP .-2
7457  1261            TAD ADDR        /LEAVE WITH FINAL ADDRESS IN AC
7460  5205            JMP LP00        /GO TO HALT FOR POSSIBLE RESTART
                / VARIABLES AND CONSTANTS
7461  0000  ADDR,     0               /LOC OF STARTING ADDRESS TO BE DUMPED
7462  0000  INDEX,    0               /COUNTER FOR NUMBER OF LOCS TO BE DUMPED
7463  0003  C3,       3               /MASK VALUES
7464  0007  C7,       7
7465  0010  C10,      10
7466  0177  C177,     177
7467  0214  C214,     214             /FORM FEED
7470  7400  C7400,    7400            /USED TO FORM SKP COMMAND
7471  7760  M20,      -20             /COUNTER FOR NUM OF BLANKS TO OUTPUT

                / OCTAL TYPEOUT ROUTINE
7472  0000  TOCT,     0
7473  7104            CLL RAL /ROTATE ADDRESS 1 LEFT
7474  3344            DCA WORD
7475  1352            TAD M4          /SET NUMBER OF DIGITS PER WORD
7476  3345            DCA NDX
7477  1344  LP03,     TAD WORD        /ROTATE WORD 3 LEFT
7500  7006            RTL
7501  7004            RAL
7502  3344            DCA WORD
7503  1344            TAD WORD
7504  0264            AND C7          /MASK BITS 9-11
7505  1351            TAD C260        /ADD 260 FOR OUTPUT
7506  4324            JMS TCHAR       /OUTPUT DIGIT
7507  2345            ISZ NDX         /DONE FOUR?
7510  5277            JMP LP03        /NO. PICK UP ANOTHER DIGIT
7511  5672            JMP I TOCT      /YES. RETURN
```

```
                / ROUTINE TO OUTPUT A CARRIAGE RETURN/LINE FEED
7512  0000  TCR,   0
7513  1347        TAD C215        /OUTPUT A C. R.
7514  4324        JMS TCHAR
7515  1346        TAD C212        /OUTPUT A L. F.
7516  4324        JMS TCHAR
7517  5712        JMP I TCR
                / ROUTINE TO OUTPUT A SPACE
7520  0000  TSP,   0
7521  1350        TAD C240        /OUTPUT A SPACE
7522  4324        JMS TCHAR
7523  5720        JMP I TSP
                / ROUTINE TO OUTPUT A CHARACTER ON TTY OR H. S. PUNCH
7524  0000  TCHAR, 0
7525  7000  SKPZ,  NOP             /CHANGED TO A "SKP" IF H. S. OUTPUT
7526  5535        JMP TCH1        /OTHERWISE GO TO TTY  OUTPUT
7527  6026        PLS
7530  7200        CLA
7531  1261        TAD ADDR        /KEEP ADDRESS IN AC WHILE PUNCHING
7532  6021        PSF
7533  5532        JMP .-1
7534  5542        JMP TCH2
7535  6046  TCH1,  TLS             /TTY OUTPUT ROUTINE
7536  7200        CLA
7537  1261        TAD ADDR
7540  6041        TSF
7541  5340        JMP .-1
7542  7200  TCH2,  CLA
7543  5724        JMP I TCHAR
                / VARIABLES AND CONSTANTS
7544  0000  WORD,  0               /STORAGE FOR DIGIT TO BE FORMATTED
7545  0000  NDX,   0               /COUNTER FOR NUM OF DIGITS OUTPUT
7546  0212  C212,  212             /CODE FOR LINE FEED
7547  0215  C215,  215             /  "  "  CARRIAGE RETURN
7550  7240  C240,  240             /  "  "  SPACE
7551  0260  C260,  260             /  "  "  FORMATTING DIGITS
7552  7774  M4,    -4              /NUMBER OF DIGITS PER WORD
```

```
AUTO    7451
C1      7455
C177    7466
C212    7546
C214    7467
C215    7547
C240    7550
C260    7551
CS      7463
C7      7464
CTRL    7470
DUMP    7400
INDEX   7462
LP00    7405
LP01    7415
LP02    7423
LP03    7477
M20     7471
M4      7552
NDX     7545
OUT     7447
SKPZ    7525
TCHAR   7524
TCH1    7535
TCH2    7542
TCR     7512
TUCT    7472
TSP     7520
WORD    7544
```